Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 4
The Cross Product

Initializations

```
> restart;
  with(VectorCalculus):
  BasisFormat(false):
>
```

The CrossProduct command

The syntax for evaluating a cross product in Maple is very simple. The following example speaks for itself.

Examples

Example 4.1.1

Compute the cross product of the vectors $\mathbf{v}_1 = \langle 1, 2, 4 \rangle$, and $\mathbf{v}_2 = \langle -5, 1, 11 \rangle$.

Solution
Use the **CrossProduct** command.

```
> v1:=<1, 2, 4>;
  v2:=<-5, 1, 11>;
```

$$v1 := \begin{bmatrix} 1 \\ 2 \\ 4 \end{bmatrix}$$

$$v2 := \begin{bmatrix} -5 \\ 1 \\ 11 \end{bmatrix} \quad (2.1.1.1)$$

```
> cp:=CrossProduct(v1, v2);
```

$$cp := \begin{bmatrix} 18 \\ -31 \\ 11 \end{bmatrix} \quad (2.1.1.2)$$

Note: For coding enthusiasts is may be nice to know that the code

$$\mathbf{v}_1 \ \&x \ \mathbf{v}_2$$

also computes the cross product of the vectors $\mathbf{v}_1$ and $\mathbf{v}_2$. The ampersand **&** indicates a noncommutative operation.After all, $\mathbf{v}_2 \times \mathbf{v}_1 = -(\mathbf{v}_1 \times \mathbf{v}_2)$.

```
> v1 &x v2;
```

$$\begin{bmatrix} 18 \\ -31 \\ 11 \end{bmatrix} \quad (2.1.1.3)$$

```
> v2 &x v1;
```

$$\begin{bmatrix} -18 \\ 31 \\ -11 \end{bmatrix} \quad (2.1.1.4)$$

4.2 Applications

Examples

Example 4.2.1

Use the cross product to compute the volume of the parallelepiped spanned by the vectors $\mathbf{a} = \langle 1, 2, 7\rangle$, $\mathbf{b} = \langle -3, 5, -1\rangle$ and $\mathbf{c} = \langle -3, 2, 2\rangle$.

Solution
As explained in class, the volume of this parallelpiped is given by the absolute value of the triple scalar product.of the three vectors.

```
> a:=<1, 2, 7>;
  b:=<-3, 5, -1>;
  c:=<-3, 2, 2>;
```

$$a := \begin{bmatrix} 1 \\ 2 \\ 7 \end{bmatrix}$$

$$b := \begin{bmatrix} -3 \\ 5 \\ -1 \end{bmatrix}$$

$$c := \begin{bmatrix} -3 \\ 2 \\ 2 \end{bmatrix} \quad (3.1.1.1)$$

```
> V:=abs( DotProduct ( CrossProduct (a, b), c));
```

$$V := 93 \quad (3.1.1.2)$$

Note: We can shorten the code significantly by using the dot • for the dot product and the

&x for the cross product.

```
> V:=abs((a &x b).c);
```

$$V := 93 \qquad (3.1.1.3)$$

```
>
```

Example 4.2.2

A bolt is tightened using a one foot wrench. A force of 20 Lb is applied to the end of the wrench under an angle of 50 degrees with the wrench vector which points from the bolt in the direction of the end of the wrench.

i) Find the magnitude of the torgue about the center of the bolt, using the formula $|\mathbf{a} \times \mathbf{b}| = |\mathbf{a}||\mathbf{b}|\sin\theta$.

ii) Compute the actual torgue vector and check the answer found under i) by computing its norm.

i) Find the magnitude of the torgue about the center of the bolt using the formula $|\mathbf{a} \times \mathbf{b}| = |\mathbf{a}||\mathbf{b}|\sin\theta$.

Solution

```
> torgue_magnitude:=simplify(1*20*sin(50*2*Pi/360));
  evalf(torgue_magnitude);
```

$$torgue\_magnitude := 20 \sin\left(\frac{5}{18}\pi\right)$$

$$15.32088886 \qquad (3.1.2.1)$$

ii) Compute the actual torgue vector and check the answer found under i) by computing its norm.

Solution

First we enter the vector representations of the wrench **r** and the force **F**. For simplicity we place both vectors in the XOY-plane and position the wrench along the positive *x*-axis.

```
> r:=<1, 0, 0>;
```

$$r := \begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix} \qquad (3.1.2.2)$$

```
> F:=20*<cos(-50*Pi/180), sin(-50*Pi/180), 0>;
```

$$F := \begin{bmatrix} 20 \cos\left(\frac{5}{18}\pi\right) \\ -20 \sin\left(\frac{5}{18}\pi\right) \\ 0 \end{bmatrix} \qquad (3.1.2.3)$$

Finally, we compute the length of this vector.

```
> test:=Norm(CrossProduct(r, F));
  evalf(test);
```

$$test := 20\sin\left(\frac{5}{18}\pi\right)$$

$$15.32088886 \qquad (3.1.2.4)$$

>

Clearly, this is the expected result.